Supplemental material for Autoregressive Latent Trajectory

Supplemental Materials for
The Longitudinal Interplay of Adolescents’ Self-Esteem and Body Image: A Conditional Autoregressive Latent Trajectory Analysis

Alexandre J.S. Morin, Ph.D.
Group for Interdisciplinary Research in Psychology Applied to Social Systems (GIRPASS), Department of Psychology, University of Sherbrooke, Canada

Christophe Maıiano, Ph.D.
UMR 6233, Institute of Movement Sciences, Etienne-Jules Marey, CNRS-University of Aix-Marseille II, Marseille, France.

Herbert W. Marsh, Ph.D.
Self-concept Enhancement and Learning Facilitation (SELF) Research Group, University of Oxford, UK

Michel Janosz, Ph.D.
School Environment Research Group, School of Psychoeducation, University of Montreal, Canada

Benjamin Nagengast, Ph.D.
Self-concept Enhancement and Learning Facilitation (SELF) Research Group, University of Oxford, UK

* The first two authors (A.J.S.M and C.M.) contributed equally to this article and their order was determined at random: both should thus be considered first authors. This manuscript was designed and partly prepared while A.J.S.M was a visiting scholar at the University of the Mediterranean - Aix-Marseille II.

Acknowledgements
This study was made possible by a grant from the Social Sciences and Humanities Research Council of Canada (SSHRC) awarded to the fourth author. The authors wish to thank Kenneth A. Bollen and Bengt O. Muthén for their invaluable advice at the initial stage of the analyses, Thierno Diallo, Paul Honey and two anonymous reviewers for their contributions to various versions of this manuscript, Kevin J. Grimm and John J. McArdle for helping us to better organize our thoughts on alternative ways of modeling change and their assumptions, as well as every student, staff member and research assistant involved in the MADDP.

Corresponding author:
Alexandre J.S. Morin
University of Sherbrooke
Department of Psychology
2500 boulevard de l’Université
Sherbrooke, QC J1K 2R1 Canada
E-mail: alexandre.morin@usherbrooke.ca
Model 1.
TITLE: Multivariate LCM model without constraints
! Text following the exclamation marks (!) is not part of the model.
DATA:
FILE IS "ALT.dat";
VARIABLE:
NAMES ARE CI SEX MIN PUB SE1 SE2 SE3 SE4 SE5 BI1 BI2 BI3 BI4 BI5;
IDVARIABLE = CI;
MISSING ARE ALL (999);
USEVARIABLES ARE SE1 SE2 SE3 SE4 SE5 BI1 BI2 BI3 BI4 BI5;
MODEL:
! To identify the intercept (i1) and slope (s1) of the body image trajectory
i1 s1 | BI1@-.4 BI2@0 BI3@1 BI4@2 BI5@3;
! To identify the intercept (i2) and slope (s2) of the self-esteem trajectory
i2 s2 | SE1@-.4 SE2@0 SE3@1 SE4@2 SE5@3;
! Intercepts and slopes are correlated by default
OUTPUT:
SAMPSTAT STANDARDIZED RESIDUAL CINTERVAL MODINDICES (3.0);
TECH1 TECH2 TECH3 TECH4;

Model 2.
TITLE: Multivariate autoregressive cross lagged model without constraints.
DATA:
FILE IS "ALT.dat";
VARIABLE:
NAMES ARE CI SEX MIN PUB SE1 SE2 SE3 SE4 SE5 BI1 BI2 BI3 BI4 BI5;
IDVARIABLE = CI;
MISSING ARE ALL (999);
USEVARIABLES ARE SE1 SE2 SE3 SE4 SE5 BI1 BI2 BI3 BI4 BI5;
MODEL:
! Autoregressive part
BI1; BI2 ON BI1; BI3 ON BI2; BI4 ON BI3; BI5 ON BI4;
SE1; SE2 ON SE1; SE3 ON SE2; SE4 ON SE3; SE5 ON SE4;
! Cross lagged part
BI2 ON SE1; BI3 ON SE2; BI4 ON SE3; BI5 ON SE4;
SE2 ON BI1; SE3 ON BI2; SE4 ON BI3; SE5 ON BI4;
! Time-specific correlations
BI2 WITH SE2; BI3 WITH SE3; BI4 WITH SE4; BI5 WITH SE5;
OUTPUT:
SAMPSTAT STANDARDIZED RESIDUAL CINTERVAL MODINDICES (3.0);
TECH1 TECH2 TECH3 TECH4;

Model 3.
TITLE: Initial multivariate ALT model without constraints
DATA:
FILE IS "ALT.dat";
VARIABLE:
NAMES ARE CI SEX MIN PUB SE1 SE2 SE3 SE4 SE5 BI1 BI2 BI3 BI4 BI5;
IDVARIABLE = CI;
MISSING ARE ALL (999);
USEVARIABLES ARE SE1 SE2 SE3 SE4 SE5 BI1 BI2 BI3 BI4 BI5;
MODEL:
! LCM part
i1 s1 | BI1@-.4 BI2@0 BI3@1 BI4@2 BI5@3;
i2 s2 | SE1@-.4 SE2@0 SE3@1 SE4@2 SE5@3;
! Correlations between the first measurement points
SE1 WITH BI1;
! Autoregressive part
BI1; BI2 ON BI1; BI3 ON BI2; BI4 ON BI3; BI5 ON BI4;
SE1; SE2 ON SE1; SE3 ON SE2; SE4 ON SE3; SE5 ON SE4;
! Cross lagged part
BI2 ON SE1; BI3 ON SE2; BI4 ON SE3; BI5 ON SE4;
SE2 ON BI1; SE3 ON BI2; SE4 ON BI3; SE5 ON BI4;
! Time-specific correlations
BI2 WITH SE2; BI3 WITH SE3; BI4 WITH SE4; BI5 WITH SE5;
OUTPUT:
SAMPSTAT STANDARDIZED RESIDUAL CINTERVAL MODINDICES (3.0);
TECH1 TECH2 TECH3 TECH4;

Model 4.
TITLE: Final multivariate ALT model with the model constraints
DATA:
FILE IS "ALT.dat";
VARIABLE:
NAMES ARE CI SEX MIN PUB SE1 SE2 SE3 SE4 SE5 BI1 BI2 BI3 BI4 BI5;
IDVARIABLE = CI;
MISSING ARE ALL (999);
USEVARIABLES ARE SE1 SE2 SE3 SE4 SE5 BI1 BI2 BI3 BI4 BI5;
MODEL:
! LCM part, with no slope on the self-esteem trajectory
i1 s1 | BI1@-.4 BI2@0 BI3@1 BI4@2 BI5@3;
i2 | SE1@-.4 SE2@0 SE3@1 SE4@2 SE5@3;
! Fixing the variance of the slope of the body image trajectory to 0
s1@0;
! Correlations between the first measurement points
SE1 WITH BI1;
! Autoregressive part, with equality constraints on body image
autoregressions
BI1; BI2 ON BI1 (1); BI3 ON BI2 (1); BI4 ON BI3 (1); BI5 ON BI4 (1);
SE1; SE2 ON SE1; SE3 ON SE2; SE4 ON SE3; SE5 ON SE4;
! Cross lagged part, with equality constraints
BI2 ON SE1 (2); BI3 ON SE2 (2); BI4 ON SE3 (2); BI5 ON SE4 (2);
SE2 ON BI1 (3); SE3 ON BI2 (3); SE4 ON BI3 (3); SE5 ON BI4 (3);
! Time-specific correlations, with equality constraints
BI2 WITH SE2 (4); BI3 WITH SE3 (4);
BI4 WITH SE4 (4); BI5 WITH SE5 (4);
OUTPUT:
SAMPSTAT STANDARDIZED RESIDUAL CINTERVAL MODINDICES (3.0);
TECH1 TECH2 TECH3 TECH4;

Model 5.
TITLE: Final conditional multivariate ALT model with the model
constraints, predictors and their interactions
DATA:
FILE IS "ALT.dat";
DEFINE:
! The DEFINE command is used to define new variables: P (pubertal development centered at the mean), PS (pubertal development by gender two-way interaction), PN (pubertal development by ethnicity two-way interaction), SN (gender by ethnicity two-way interaction), PSN (three-way interaction). The same command can be used to recode the variables to calculate the simple slopes of the interactions. For example "sexe = sex - 1" is use to invert gender.
P = PUB - 2.4729;
PS = P * Sex; PN = P * Min; SN = Sex * Min;
PSN = P * Sex * Min;
VARIABLE:
NAMES ARE CI SEX MIN PUB SE1 SE2 SE3 SE4 SE5 BI1 BI2 BI3 BI4 BI5;
IDVARIABLE = CI;
MISSING ARE ALL (999);
! The “DEFINE” variables are added at the end of the list.
USEVARIABLES ARE SEX MIN SE1 SE2 SE3 SE4 SE5 BI1 BI2 BI3 BI4 BI5 P PS PN PSN;
MODEL:
! ALT, as in the previous unconditional model
i1 | BI1@-.4 BI2@0 BI3@1 BI4@2 BI5@3;
i2 | SE1@-.4 SE2@0 SE3@1 SE4@2 SE5@3;
s1@0;
SE1 WITH BI1;
BI1; BI2 ON BI1 (1); BI3 ON BI2 (1); BI4 ON BI3 (1); BI5 ON BI4 (1);
SE1; SE2 ON SE1; SE3 ON SE2; SE4 ON SE3; SE5 ON SE4;
BI2 ON SE1 (2); BI3 ON SE2 (2); BI4 ON SE3 (2); BI5 ON SE4 (2);
SE2 ON BI1 (3); SE3 ON BI2 (3); SE4 ON BI3 (3); SE5 ON BI4 (3);
BI2 WITH SE2 (4); BI3 WITH SE3 (4);
BI4 WITH SE4 (4); BI5 WITH SE5 (4);
! Additional parts for the conditional model
! intercepts, slopes and first-point correlations need to be specified.
SE1 i1 i2 WITH BI1;
SE1 i1 WITH i2;
! Predictive part of the model
i1 s1 i2 SE1 BI1 ON SEX MIN P PS PM SM PSM;
OUTPUT:
SAMPSTAT STANDARDIZED RESIDUAL CINTERVAL MODINDICES (3.0);
TECH1 TECH2 TECH3 TECH4;
Section 2.

Results from the unconditional univariate models.

The results from the different univariate models of self-esteem and body image are reported in the supplemental Table. For self-esteem, the results show that the autoregressive model does not provide an optimal degree of fit to the data, while the ALT and LCM provide comparable and adequate levels of fit to the data. However, the ALT proved superior to the nested ALT-LCM according the $\Delta \chi^2$ statistic. The results from the following ALT models suggest that the slope factor can be removed (model 6) without significantly changing the overall fit of the model (i.e. that self-esteem trajectories are stable), but that the imposition of equality constraints on the autoregressive parameters (model 7) result in a significantly poorer fit. This suggests that the ability of self-esteem to predict later levels of self-esteem increases slightly throughout adolescence, although it remain small once the overall trajectories are taken into account [(\(\rho_{t2, t1} = 0.21; \text{s.e.} = 0.03\)); (\(\rho_{t4, t2} = 0.21; \text{s.e.} = 0.03\)); (\(\rho_{t6, t5} = 0.24; \text{s.e.} = 0.03\)]. Further examination of these results reveal that the mean and variance of the intercept factor are positive and significant, suggesting an elevated and stable level of self-esteem but significant inter-individual variability between the students [(\(\mu_{t1} = 31.73; \text{s.e.} = 0.18\)); (\(\mu_{a} = 25.03; \text{s.e.} = 0.89\)); (\(\psi_{t1 t1} = 30.51; \text{s.e.} = 1.37\)); (\(\psi_{aa} = 9.36; \text{s.e.} =1.18\)]. The observable difference between Time 1 self-esteem (\(\mu_{t1}\)) and the intercept (\(\mu_{a}\)) stems from the fact that in ALTs, the intercept represents the portion of Time-2 self-esteem remaining unexplained by Time-1 self-esteem. In fact, if one were to multiply \(\mu_{t1} (31.73)\) by the autoregressive parameter (0.21) and add this to \(\mu_{a} (25.03)\), the result (31.69) would represent the model estimated level self-esteem at Time-2. It should be noted that the covariance between the first measurement point and the intercept factor is also elevated and significant (\(\psi_{t1 a} = 10.15; \text{s.e.} = 0.92\)), confirming self-esteem developmental stability.

Regarding body image, the results show that the autoregressive model does not provide an optimal degree of fit to the data, while the ALT and LCM provide comparable and adequate levels of fit to the data. However, the ALT once again proved superior to the nested ALT-LCM according the $\Delta \chi^2$ statistic. The complete univariate ALT model for body image resulted in the estimation of a
negative slope variance, suggesting that inter-individual variability on the developmental changes in body image across adolescence was negligible. Thus, this model was directly re-estimated by fixing the variance of the slope factor to zero (model 5). The resulting model provides an adequate level of fit to the data according to most fit indices, with the exception of the RMSEA (.081) and fit the data significantly better than the comparative nested ALT-LCM. The results from the following ALT models suggest that the slope factor should not be removed (i.e. that adolescents body image levels evolve over time; model 6), but that equality constraints may be imposed on the autoregressive parameters (model 7). Meaning that the ability of body image to predict later levels of body image remain stable and moderate in adolescence ($\rho_{t, t-1} = 0.37$; s.e. = 0.03). The results also suggest that, although significant inter-individual variability is present on the initially high levels of body image [(μ$_{t1}$ = 23.42; s.e. = 0.18); (μ$_{a}$ = 14.42; s.e. = 0.32); ($\psi_{tt1}$ = 32.83; s.e. = 1.48); ($\psi_{aa}$ = 4.21; s.e. = 0.80)], the slight increase observed in body image ($\mu_{b} = 0.32; $ s.e. = 0.06) appears common to all participants. It should be noted that the covariance between the first measurement point and the intercept factor is also elevated and significant ($\psi_{t1a} = 7.84; $ s.e. = 0.85).
### Supplemental Table

Results from the Unconditional Univariate Latent Curve Models (LCM), Autoregressive Models and Autoregressive Latent Trajectory (ALT) Models.

<table>
<thead>
<tr>
<th></th>
<th>$\chi^2$ (df)</th>
<th>CM</th>
<th>$\Delta \chi^2$ (df)</th>
<th>CFI</th>
<th>TLI</th>
<th>RMSEA</th>
<th>SRMR</th>
</tr>
</thead>
<tbody>
<tr>
<td><strong>Self Esteem</strong></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>1- LCM, full model</td>
<td>58.13 (10)*</td>
<td>---</td>
<td>---</td>
<td>0.966</td>
<td>0.966</td>
<td>0.069</td>
<td>0.090</td>
</tr>
<tr>
<td>2- Autoregressive, full model</td>
<td>191.58 (6)*</td>
<td>---</td>
<td>---</td>
<td>0.870</td>
<td>0.784</td>
<td>0.175</td>
<td>0.121</td>
</tr>
<tr>
<td>3- ALT, full model</td>
<td>19.18 (3)*</td>
<td>---</td>
<td>---</td>
<td>0.989</td>
<td>0.962</td>
<td>0.073</td>
<td>0.055</td>
</tr>
<tr>
<td>4- ALT, nested LCM model</td>
<td>40.27 (7)*</td>
<td>3</td>
<td>21.09 (4)*</td>
<td>0.977</td>
<td>0.967</td>
<td>0.069</td>
<td>0.053</td>
</tr>
<tr>
<td>5- ALT, no slope variance</td>
<td>25.70 (6)*</td>
<td>3</td>
<td>6.52 (3)</td>
<td>0.986</td>
<td>0.977</td>
<td>0.057</td>
<td>0.055</td>
</tr>
<tr>
<td>6- ALT, no slope</td>
<td>26.34 (7)*</td>
<td>5</td>
<td>0.64 (1)</td>
<td>0.986</td>
<td>0.981</td>
<td>0.052</td>
<td>0.055</td>
</tr>
<tr>
<td>7- ALT-6 + fixed regressions</td>
<td>52.31 (10)*</td>
<td>6</td>
<td>25.97 (3)*</td>
<td>0.970</td>
<td>0.970</td>
<td>0.065</td>
<td>0.055</td>
</tr>
<tr>
<td><strong>Body Image</strong></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>1- LCM, full model</td>
<td>76.75 (10)*</td>
<td>---</td>
<td>---</td>
<td>0.968</td>
<td>0.968</td>
<td>0.081</td>
<td>0.072</td>
</tr>
<tr>
<td>2- Autoregressive, full model</td>
<td>144.77 (6)*</td>
<td>---</td>
<td>---</td>
<td>0.933</td>
<td>0.888</td>
<td>0.152</td>
<td>0.096</td>
</tr>
<tr>
<td>3- ALT, full model</td>
<td>22.65 (3)</td>
<td>---</td>
<td>---</td>
<td>0.991</td>
<td>0.968</td>
<td>0.081</td>
<td>0.054</td>
</tr>
<tr>
<td>4- ALT, nested LCM model</td>
<td>61.98 (7)*</td>
<td>3</td>
<td>39.33 (4)*</td>
<td>0.973</td>
<td>0.962</td>
<td>0.088</td>
<td>0.060</td>
</tr>
<tr>
<td>5- ALT, no slope variance</td>
<td>42.53 (6)*</td>
<td>3</td>
<td>19.88 (3)*</td>
<td>0.982</td>
<td>0.971</td>
<td>0.078</td>
<td>0.061</td>
</tr>
<tr>
<td>6- ALT, no slope</td>
<td>53.42 (7)*</td>
<td>5</td>
<td>10.89 (1)*</td>
<td>0.978</td>
<td>0.968</td>
<td>0.081</td>
<td>0.078</td>
</tr>
<tr>
<td>7- ALT-5 + fixed regressions</td>
<td>49.49 (9)*</td>
<td>5</td>
<td>6.96 (3)</td>
<td>0.980</td>
<td>0.978</td>
<td>0.067</td>
<td>0.068</td>
</tr>
</tbody>
</table>

*Notes:* $\chi^2$ = chi-square test of model fit; df = degrees of freedom; $\Delta \chi^2$ = chi square difference test; CM = comparison model in the $\Delta \chi^2$; CFI = comparative fit index; TLI = Tucker-Lewis index; RMSEA = root mean square error of approximation; SRMR = standardized root mean square residual; EP = model retained given estimation problems (negative or non-significant variances) in the previous one.

*p ≤ 0.01